NAG Fortran Library Routine Document

F07APF (ZGESVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07APF (ZGESVX) uses the LU factorization to compute the solution to a complex system of linear equations

$$AX = B \quad \text{or} \quad A^TX = B \quad \text{or} \quad A^HX = B$$

where A is an n by n matrix and X and B are n by r matrices. Error bounds on the solution and a condition estimate are also provided.

2 Specification

```
 SUBROUTINE F07APF (FACT, TRANS, N, NRHS, A, LDA, AF, LDAF, IPIV, EQUED,
1                   R, C, B, LDB, X, LDX, RCOND, FERR, BERR, WORK, RWORK,
2                   INFO)
 INTEGER            N, NRHS, LDA, LDAF, IPIV(*), LDB, LDX, INFO
 double precision   R(*), C(*), RCOND, FERR(*), BERR(*), RWORK(*)
 complex*16         A(LDA,*), AF(LDAF,*), B(LDB,*), X(LDX,*), WORK(*)
 CHARACTER*1        FACT, TRANS, EQUED
```

The routine may be called by its LAPACK name ***zgesvx***.

3 Description

The following steps are performed:

1. If FACT = 'E', real scaling factors are computed to equilibrate the system:

   if TRANS = 'N', $(D_RAD_C)(D_C^{-1}) = D_RB$;
   if TRANS = 'T', $(D_RAD_C)^T(D_R^{-1}X) = D_CB$;
   if TRANS = 'C', $(D_RAD_C)^H(D_R^{-1}X) = D_CB$;

   where D_R and D_C are diagonal matrices with positive diagonal elements.

   Whether or not the system will be equilibrated depends on the scaling of the matrix A, but if equilibration is used, A is overwritten by D_RAD_C and B by D_RB (if TRANS = 'N') or D_CB (if TRANS = 'T' or 'C').

2. If FACT = 'N' or 'E', the LU decomposition is used to factor the matrix A (after equilibration if FACT = 'E') as

   $$A = PLU,$$

   where P is a permutation matrix, L is a unit lower triangular matrix, and U is upper triangular.

3. If some $u_{ii} = 0$, so that U is exactly singular, then the routine returns with INFO $= i$. Otherwise, the factorized form of A is used to estimate the condition number of the matrix A. If the reciprocal of the condition number is less than ***machine precision***, INFO $=$ N $+ 1$ is returned as a warning, but the routine still goes on to solve for X and compute error bounds as described below.

4. The system of equations is solved for X using the factorized form of A.

5. Iterative refinement is applied to improve the computed solution matrix and to calculate error bounds and backward error estimates for the computed solution.
6. If equilibration was used, the matrix X is premultiplied by D_C (if TRANS = 'N') or D_R (if TRANS = 'T' or 'C') so that it solves the original system before equilibration.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

1: FACT – CHARACTER*1 *Input*

On entry: specifies whether or not the factored form of the matrix A is supplied on entry, and if not, whether the matrix A should be equilibrated before it is factored:

if FACT = 'F' on entry, AF and IPIV contain the factored form of A. If EQUED $\neq$ 'N', the matrix A has been equilibrated with scaling factors given by R and C. A, AF and IPIV are not modified;
if FACT = 'N', the matrix A will be copied to AF and factored;
if FACT = 'E', the matrix A will be equilibrated if necessary, then copied to AF and factored.

Constraint: FACT = 'F', 'N' or 'E'.

2: TRANS – CHARACTER*1 *Input*

On entry: specifies the form of the system of equations:

if TRANS = 'N', $AX = B$ (No transpose);
if TRANS = 'T', $A^T X = B$ (Transpose);
if TRANS = 'C', $A^H X = B$ (Conjugate transpose).

Constraint: TRANS = 'N', 'T' or 'C'.

3: N – INTEGER *Input*

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: N $\geq$ 0.

4: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS $\geq$ 0.

5: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n matrix A.

If FACT = 'F' and EQUED $\neq$ 'N', A must have been equilibrated by the scaling factors in R and/or C.

On exit: if EQUED $\neq$ 'N', A is scaled as follows:

if EQUED = 'R', A = $D_R A$;
if EQUED = 'C', A = $A D_C$;
if EQUED = 'B', A = $D_R A D_C$.

A is not modified if FACT = 'F' or 'N', or if FACT = 'E' and EQUED = 'N' on exit.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07APF (ZGESVX) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

7: AF(LDAF,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array AF must be at least $\max(1, \mathrm{N})$.

On entry: if FACT = 'F', AF contains the factors L and U from the factorization $A = PLU$ as computed by F07ARF (ZGETRF).

If EQUED $\neq$ 'N', AF is the factored form of the equilibrated matrix A.

If FACT = 'N' or 'E', AF need not be set.

On exit: if FACT = 'N', AF returns the factors L and U from the factorization $A = PLU$ of the original matrix A.

If FACT = 'E', AF returns the factors L and U from the factorization $A = PLU$ of the equilibrated matrix A (see the description of A for the form of the equilibrated matrix).

If FACT = 'F', AF is unchanged from entry.

8: LDAF – INTEGER *Input*

On entry: the first dimension of the array AF as declared in the (sub)program from which F07APF (ZGESVX) is called.

Constraint: LDAF $\geq \max(1, \mathrm{N})$.

9: IPIV($*$) – INTEGER array *Input/Output*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: if FACT = 'F', IPIV contains the pivot indices from the factorization $A = PLU$ as computed by F07ARF (ZGETRF); at the ith step row i of the matrix was interchanged with row IPIV(i).

If FACT = 'N' or 'E', IPIV need not be set. IPIV$(i) = i$ indicates a row interchange was not required.

On exit: if FACT = 'N', IPIV contains the pivot indices from the factorization $A = PLU$ of the original matrix A.

If FACT = 'E', IPIV contains the pivot indices from the factorization $A = PLU$ of the equilibrated matrix A.

If FACT = 'E', IPIV is unchanged from entry.

10: EQUED – CHARACTER*1 *Input/Output*

On entry: if FACT = 'N' or 'E', EQUED need not be set.

If FACT = 'F', EQUED must specify the form of the equilibration that was performed as follows:

if EQUED = 'N', no equilibration;
if EQUED = 'R', row equilibration, i.e., A has been premultiplied by D_R;
if EQUED = 'C', column equilibration, i.e., A has been postmultiplied by D_C;
if EQUED = 'B', both row and column equilibration, i.e., A has been replaced by $D_R A D_C$.

On exit: if FACT = 'F', EQUED is unchanged from entry.

Otherwise, if INFO $\geq$ 0, EQUED specifies the form of equilibration that was performed as specified above.

Constraint: if FACT = 'F' , EQUED = 'N', 'R', 'C' or 'B'.

11: R($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array R must be at least $\max(1, \mathrm{N})$.

On entry: if FACT = 'N' or 'E', R need not be set.

If FACT = 'F' and EQUED = 'R' or 'B', R must contain the row scale factors for A, D_R; each element of R must be positive.

On exit: if FACT = 'F', R is unchanged from entry.

Otherwise, if INFO $\geq$ 0 and EQUED = 'R' or 'B', R contains the row scale factors for A, D_R, such that A is multiplied on the left by D_R; each element of R is positive.

12: C($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array C must be at least $\max(1, \mathrm{N})$.

On entry: if FACT = 'N' or 'E', C need not be set.

If FACT = 'F' or EQUED = 'C' or 'B', C must contain the column scale factors for A, D_C; each element of C must be positive.

On exit: if FACT = 'F', C is unchanged from entry.

Otherwise, if INFO $\geq$ 0 and EQUED = 'C' or 'B', C contains the row scale factors for A, D_C; each element of C is positive.

13: B(LDB,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: if EQUED = 'N', B is not modified.

If TRANS = 'N' and EQUED = 'R' or 'B', B is overwritten by $D_R B$.

If TRANS = 'T' or 'C' and EQUED = 'C' or 'B', B is overwritten by $D_C B$.

14: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07APF (ZGESVX) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

15: X(LDX,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array X must be at least $\max(1, \mathrm{NRHS})$.

On exit: if INFO = 0 or INFO = N + 1, the n by r solution matrix X to the original system of equations. Note that the arrays A and B are modified on exit if EQUED $\neq$ 'N', and the solution to the equilibrated system is $D_C^{-1}X$ if TRANS = 'N' and EQUED = 'C' or 'B', or $D_R^{-1}X$ if TRANS = 'T' or 'C' and EQUED = 'R' or 'B'.

16: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which F07APF (ZGESVX) is called.

Constraint: LDX $\geq \max(1, \mathrm{N})$.

17: RCOND – ***double precision*** *Output*

On exit: if INFO ≥ 0, an estimate of the reciprocal condition number of the matrix A (after equilibration if that is performed), computed as $\text{RCOND} = 1/\left(\|A\|_1 \|A^{-1}\|_1\right)$.

18: FERR($*$) – ***double precision*** array *Output*

Note: the dimension of the array FERR must be at least $\max(1, \text{NRHS})$.

On exit: if INFO $= 0$ or INFO $= \text{N} + 1$, an estimate of the forward error bound for each computed solution vector, such that $\|\hat{x}_j - x_j\|_\infty / \|x_j\|_\infty \leq \text{FERR}(j)$ where $\hat{x}_j$ is the jth column of the computed solution returned in the array X and x_j is the corresponding column of the exact solution X. The estimate is as reliable as the estimate for RCOND, and is almost always a slight overestimate of the true error.

19: BERR($*$) – ***double precision*** array *Output*

Note: the dimension of the array BERR must be at least $\max(1, \text{NRHS})$.

On exit: if INFO $= 0$ or INFO $= \text{N} + 1$, an estimate of the componentwise relative backward error of each computed solution vector $\hat{x}_j$ (i.e., the smallest relative change in any element of A or B that makes $\hat{x}_j$ an exact solution).

20: WORK($*$) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 2 \times \text{N})$.

21: RWORK($*$) – ***double precision*** array *Output*

Note: the dimension of the array RWORK must be at least $\max(1, 2 \times \text{N})$.

On exit: $\text{RWORK}(1)$ contains the reciprocal pivot growth factor $\|A\|/\|U\|$. The 'max absolute element' norm is used. If $\text{RWORK}(1)$ is much less than 1, then the stability of the LU factorization of the (equilibrated) matrix A could be poor. This also means that the solution X, condition estimator RCOND, and forward error bound FERR could be unreliable. If factorization fails with $0 <$ INFO $\leq$ N, then $\text{RWORK}(1)$ contains the reciprocal pivot growth factor for the leading INFO columns of A.

22: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO $= \text{N} + 1$

U is nonsingular, but RCOND is less than ***machine precision***, so that the matrix A is numerically singular. A solution to the equations $AX = B$, and corresponding error bounds, have nevertheless been computed because there are some situations where the computed solution can be more accurate that the value of RCOND would suggest.

INFO > 0 and INFO $\leq$ N

If INFO $= i$, u_{ii} is exactly zero. The factorization has been completed, but the factor U is exactly singular, so the solution and error bounds could not be computed. $\text{RCOND} = 0$ is returned.

U is nonsingular, but RCOND is less than ***machine precision***, meaning that the matrix is singular to working precision. Nevertheless, the solution and error bounds are computed because there are a number of situations where the computed solution can be more accurate than the value of RCOND would suggest.

7 Accuracy

For each right-hand side vector b, the computed solution $\hat{x}$ is the exact solution of a perturbed system of equations $(A+E)\hat{x}=b$, where

$$|E| \le c(n)\epsilon P|L||U|,$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***. See Section 9.3 of Higham (2002) for further details.

If x is the true solution, then the computed solution $\hat{x}$ satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|\hat{x}\|_\infty} \le w_c \operatorname{cond}(A,\hat{x},b)$$

where $\operatorname{cond}(A,\hat{x},b) = \big\||A^{-1}|(|A||\hat{x}|+|b|)\big\|_\infty / \|\hat{x}\|_\infty \le \operatorname{cond}(A) = \big\||A^{-1}||A|\big\|_\infty \le \kappa_\infty(A)$. If $\hat{x}$ is the jth column of X, then w_c is returned in $\mathrm{BERR}(j)$ and a bound on $\|x-\hat{x}\|_\infty/\|\hat{x}\|_\infty$ is returned in $\mathrm{FERR}(j)$. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Further Comments

The factorization of A requires approximately $\frac{8}{3}n^3$ floating point operations.

Estimating the forward error involves solving a number of systems of linear equations of the form $Ax=b$ or $A^Tx=b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $8n^2$ operations.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The real analogue of this routine is F07ABF (DGESVX).

9 Example

To solve the equations

$$AX = B,$$

where A is the general matrix

$$A = \begin{pmatrix} -1.34+2.55i & 0.28+3.17i & -6.39-2.20i & 0.72-0.92i \\ -1.70-14.10i & 33.10-1.50i & -1.50+13.40i & 12.90+13.80i \\ -3.29-2.39i & -1.91+4.42i & -0.14-1.35i & 1.72+1.35i \\ 2.41+0.39i & -0.56+1.47i & -0.83-0.69i & -1.96+0.67i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 26.26+51.78i & 31.32-6.70i \\ 64.30-86.80i & 158.60-14.20i \\ -5.75+25.31i & -2.15+30.19i \\ 1.16+2.57i & -2.56+7.55i \end{pmatrix}.$$

Error estimates for the solutions, information on scaling, an estimate of the reciprocal of the condition number of the scaled matrix A and an estimate of the reciprocal of the pivot growth factor for the factorization of A are also output.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07APF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
      INTEGER          LDA, LDAF, LDB, LDX, NRHSMX
      PARAMETER        (LDA=NMAX,LDAF=NMAX,LDB=NMAX,LDX=NMAX,
     +                 NRHSMX=NMAX)
*     .. Local Scalars ..
      DOUBLE PRECISION RCOND
      INTEGER          I, IFAIL, INFO, J, N, NRHS
      CHARACTER        EQUED
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), AF(LDAF,NMAX), B(LDB,NRHSMX),
     +                 WORK(2*NMAX), X(LDX,NRHSMX)
      DOUBLE PRECISION BERR(NRHSMX), C(NMAX), FERR(NRHSMX), R(NMAX),
     +                 RWORK(2*NMAX)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         X04DBF, ZGESVX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07APF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHSMX) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Solve the equations AX = B for X
*
         CALL ZGESVX('Equilibrate','No transpose',N,NRHS,A,LDA,AF,LDAF,
     +               IPIV,EQUED,R,C,B,LDB,X,LDX,RCOND,FERR,BERR,WORK,
     +               RWORK,INFO)
*
         IF ((INFO.EQ.0) .OR. (INFO.EQ.N+1)) THEN
*
*           Print solution, error bounds, condition number, the form
*           of equilibration and the pivot growth factor
*
            IFAIL = 0
            CALL X04DBF('General',' ',N,NRHS,X,LDX,'Bracketed','F7.4',
     +                  'Solution(s)','Integer',RLABS,'Integer',CLABS,
     +                  80,0,IFAIL)
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Backward errors (machine-dependent)'
            WRITE (NOUT,99999) (BERR(J),J=1,NRHS)
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Estimated forward error bounds (machine-dependent)'
            WRITE (NOUT,99999) (FERR(J),J=1,NRHS)
            WRITE (NOUT,*)
            IF (EQUED.EQ.'N') THEN
               WRITE (NOUT,*) 'A has not been equilibrated'
            ELSE IF (EQUED.EQ.'R') THEN
               WRITE (NOUT,*) 'A has been row scaled as diag(R)*A'
            ELSE IF (EQUED.EQ.'C') THEN
```

```
               WRITE (NOUT,*) 'A has been column scaled as A*diag(C)'
            ELSE IF (EQUED.EQ.'B') THEN
               WRITE (NOUT,*)
     +           'A has been row and column scaled as diag(R)*A*diag(C)'
            END IF
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Reciprocal condition number estimate of scaled matrix'
            WRITE (NOUT,99999) RCOND
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Estimate of reciprocal pivot growth factor'
            WRITE (NOUT,99999) RWORK(1)
*
            IF (INFO.EQ.N+1) THEN
               WRITE (NOUT,*)
               WRITE (NOUT,*)
     +           'The matrix A is singular to working precision'
            END IF
         ELSE
            WRITE (NOUT,99998) 'The (', INFO, ',', INFO, ')',
     +        ' element of the factor U is zero'
         END IF
      ELSE
         WRITE (NOUT,*) 'NMAX and/or NRHSMX too small'
      END IF
      STOP
*
99999 FORMAT ((3X,1P,7E11.1))
99998 FORMAT (1X,A,I3,A,I3,A,A)
      END
```

9.2 Program Data

```
F07APF Example Program Data

   4               2                                   :Values of N and NRHS

 (-1.34,  2.55) (  0.28,  3.17) (-6.39,-2.20) ( 0.72,-0.92)
 (-1.70,-14.10) ( 33.10, -1.50) (-1.50,13.40) (12.90,13.80)
 (-3.29, -2.39) ( -1.91,  4.42) (-0.14,-1.35) ( 1.72, 1.35)
 ( 2.41,  0.39) ( -0.56,  1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A

 (26.26, 51.78) ( 31.32, -6.70)
 (64.30,-86.80) (158.60,-14.20)
 (-5.75, 25.31) ( -2.15, 30.19)
 ( 1.16,  2.57) ( -2.56,  7.55)                             :End of matrix B
```

9.3 Program Results

```
F07APF Example Program Results

Solution(s)
                   1                 2
1  ( 1.0000, 1.0000) (-1.0000,-2.0000)
2  ( 2.0000,-3.0000) ( 5.0000, 1.0000)
3  (-4.0000,-5.0000) (-3.0000, 4.0000)
4  ( 0.0000, 6.0000) ( 2.0000,-3.0000)

Backward errors (machine-dependent)
      3.5E-17    1.0E-16

Estimated forward error bounds (machine-dependent)
      5.6E-14    8.0E-14

A has been row scaled as diag(R)*A

Reciprocal condition number estimate of scaled matrix
      1.0E-02

Estimate of reciprocal pivot growth factor
      8.3E-01
```